Experimenting with Uncertainty

Consider a single qubit in a given state $|\psi\rangle$. We perform measurements with the spin-1/2 operators

$$S_x = \frac{\hbar}{2}\begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}, \text{ and } S_y = \frac{\hbar}{2}\begin{pmatrix} 0 & -i \\ i & 0 \end{pmatrix}$$

Define an arbitrary qubit state vector

In[43]:=
```
ClearAll["Global`*"]
θ = Random[Real, {0, 2 Pi}];
ϕ = Random[Real, {0, 2 Pi}];
β = Random[Real, {0, 2 Pi}];
```

In[47]:=
```
psi = Exp[I β] {Cos[θ / 2], Exp[I ϕ] Sin[θ / 2]}
```

Out[47]=
```
{0.0940971 + 0.722775 ⅈ, -0.671955 - 0.131218 ⅈ}
```

We use the Born rule to calculate the probability that a measurement with S_x yields $\pm\hbar/2$ (below we set $\hbar$=1, in order to simplify the arithmetic)

In[48]:=
```
ℏ = 1;
```

Eigenstate for $S_x = \hbar/2$

In[49]:=
```
psixplus = {1, 1} / Sqrt[2];
```

Eigenstate for $S_x = -\hbar/2$

In[50]:=
```
psixminus = {1, -1} / Sqrt[2];

pxplus = Abs[psixplus.psi] ^2;
pxminus = Abs[psixminus.psi] ^2;
pxplus + pxminus == 1
```

Out[54]=
```
True
```

Define measurement device S_x

In[55]:=
```
SXmeasurement := RandomChoice[{pxplus, pxminus} -> {ℏ / 2, -ℏ / 2}];
```

Let's take a set of 10 independent measurements with S_x (assuming for each measurement the system is in state $|\psi\rangle$).

In[56]:=
```
Table[SXmeasurement, {i, 1, 10}]
```

Out[56]= $\left\{\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, \frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, \frac{1}{2}\right\}$

As expected, because state $|\psi\rangle$ is not necessarily an eigenstate of S_x, and the result of a measurement is probabilistic. In order to attain a higher quality of statistics we need to make more that just 10 measurements. Lets construct a sample space with 100 elements,

In[57]:=
```
sample = 100;
```

In[58]:=
```
Sxsamples = Table[SXmeasurement, {i, 1, sample}];
```

Let's take the average, or mean, value $\langle S_x \rangle$

In[59]:=
```
? Mean
```

Mean[*list*] gives the statistical mean of the elements in *list*.
Mean[*dist*] gives the mean of the distribution *dist*. >>

In[60]:=
```
xmean = Mean[Sxsamples]
```

Out[60]= $-\frac{21}{100}$

We also find the variance of this distribution

In[61]:=
```
? Variance
```

Variance[*list*] gives the sample variance of the elements in *list*.
Variance[*dist*] gives the variance of the distribution *dist*. >>

In[62]:=
```
varx = Simplify[Variance[Sxsamples], Assumptions → ℏ ∈ Reals]
```

Out[62]= $\frac{2059}{9900}$

By definition, the uncertainty (the square-root of the variance), can be calculated using the relation

$$\Delta S_x \equiv \sqrt{\langle S_x{}^2 \rangle - \langle S_x \rangle^2}$$

But the Mathematica command *Variance* evaluates it directly from the distribution of values

In[63]:=
```
deltaSx = Sqrt[varx]
```

Out[63]= $\frac{\sqrt{\frac{2059}{11}}}{30}$

Now repeat the above steps to calculate ΔS_y
Eigenstate for $S_y = \hbar/2$

In[64]:=
```
psiyplus = {1, I} / Sqrt[2];
```

Eigenstate for $S_y = -\hbar/2$

In[65]:=
```
psiyminus = {1, -I} / Sqrt[2];

pyplus = Abs[Conjugate[psiyplus].psi] ^ 2;
pyminus = Abs[Conjugate[psiyminus].psi] ^ 2;
pyplus + pyminus == 1
```
Out[69]= True

In[70]:=
```
SYmeasurement := RandomChoice[{pyplus, pyminus} -> {ħ / 2, -ħ / 2}];
```
In[71]:=
```
Sysamples = Table[SYmeasurement, {i, 1, sample}];
```
In[72]:=
```
vary = Simplify[Variance[Sysamples], Assumptions → ħ ∈ Reals]
```
Out[72]= $\frac{49}{2475}$

In[73]:=
```
deltaSy = Sqrt[vary]
```
Out[73]= $\frac{7}{15\sqrt{11}}$

Finally, in order to check the validity of the uncertainty relations

$$\Delta S_x \ \Delta S_y \geq 1/2 \left\langle [S_x, S_y] \right\rangle = \hbar/2 \langle S_z \rangle$$

We need to evaluate $\langle S_z \rangle$

Eigenstate for $S_z = \hbar/2$

In[74]:=
```
psizplus = {1, 0};
```
Eigenstate for $S_z = -\hbar/2$

In[75]:=
```
psizminus = {0, 1};
```
In[76]:=
```
pzplus = Abs[psizplus.psi] ^ 2;
pzminus = Abs[psizminus.psi] ^ 2;
pzplus + pzminus == 1
```
Out[79]= True

In[80]:=
```
SZmeasurement := RandomChoice[{pzplus, pzminus} -> {ħ / 2, -ħ / 2}];
```
In[81]:=
```
zsamples = Table[SZmeasurement, {i, 1, sample}];
```
Evaluate the left hand side of the uncertainty relation

In[82]:=
```
lefthandside = N[deltaSx deltaSy / ħ ^ 2 ]
```
Out[82]= 0.0641684

Evaluate the right hand side of the uncertainty relation

In[83]:=
```
righthandside = N[Abs[Mean[zsamples] / ℏ / 2]]
```

Out[83]=
```
0.005
```

Check if the uncertainty relation holds

In[84]:=
```
lefthandside > righthandside
```

Out[84]=
```
True
```

Exercises

(1) Repeat the calculations for various states $|\psi\rangle$, chosen, as above randomly. For every experiment you perform, does the above equality always hold. If not, why ? If you find the latter does that invalidate the uncertainty relation ? (Hint: try using a larger value for the sample size)

(2) Instead of operators S_x,S_y perform measurements with the operator $S_x\, S_y$ and $S_y\, S_x$. Derive an uncertainty relation for this pair of measurements.